Tilting A Truncated Icosahedron

Finding The Maximum Tilt Angle And The Associated Liquid Volume

John Snyder
June, 2016

Problem

The marketing boys behind a large beverage brand decide to take advantage of a much-hyped soccer tournament by changing the shape of their plastic canisters. The new canisters are shaped as a truncated icosahedra (achieved by taking an icosahedron and cutting off one-third of each edge on both ends; see http://mathworld.wolfram.com/TruncatedIcosahedron.html), which are often associated with the sport. The canisters have a small tap just above the bottom face. A pentagon was chosen as the bottom, as this accentuates the symmetry of the shape. There were, however, some concerns about the stability of the canister, which is typically placed near the edge of a table top.

On the assumption that the empty weight of the canister is negligible and its walls are infinitely thin, what is the maximum tilt angle of the table, so that the canister will never lose its balance, independent of the direction of the tilt and the filling level of the canister? If the maximum tilt angle is marginally exceeded and the canister starts to tumble, what fraction of its volume is filled with fluid?

Solution

Summary Of Results

The maximum tilt angle of the table so that the canister will never lose its balance, independent of the direction of the tilt and the filling level of the canister, is 16.1787°, this being 0.282372 radians. When this angle is marginally exceeded and the canister starts to tumble, the fraction of its volume which is filled with fluid is 11.7396%.

Here is a visualization of the canister when it is in this optimal state. We see the truncated icosahedron, the liquid it contains, the tilted table top, and a green pentagon representing the position of the base of the polyhedron in space before the table was tilted.



Preliminaries

We will solve the problem using *Mathematica*. Here is a truncated icosahedron:

In[1]:=
```
Show[PolyhedronData["TruncatedIcosahedron"], Boxed → False]
```

Out[1]=



All edges are of length 1.

In[2]:=
```
PolyhedronData["TruncatedIcosahedron", "EdgeLengths"]
```

Out[2]= {1}

Its volume is:

In[3]:=
```
PolyhedronData["TruncatedIcosahedron", "Volume"]
```

Out[3]= $\frac{1}{4}\left(125 + 43\sqrt{5}\right)$

In[4]:=
```
N[%]
```

Out[4]= 55.2877

It has 90 edges and 60 vertices.

In[5]:=
```
PolyhedronData["TruncatedIcosahedron", "EdgeCount"]
```

Out[5]= 90

In[6]:=
```
PolyhedronData["TruncatedIcosahedron", "VertexCount"]
```

Out[6]= 60

It is centered at the origin, this being its centroid (center of mass).

In[7]:=
```
PolyhedronData["TruncatedIcosahedron", "Centroid"]
```

Out[7]= {0, 0, 0}

We get the exact coordinates of all the vertices in the next cell, but we display only an abbreviated list.

In[8]:=
```
vcor = PolyhedronData["TruncatedIcosahedron", "VertexCoordinates"];
Short[vcor, 2]
```

Out[9]//Short=
$$\left\{\left\{-\frac{1}{2}\sqrt{1-\frac{2}{\sqrt{5}}},\ -1-\frac{\sqrt{5}}{2},\ \sqrt{\frac{9}{8}+\frac{9}{8\sqrt{5}}}\right\},\ \ll 58 \gg,\ \left\{\text{Root}\left[1-5\,\#1^2+5\,\#1^4\,\&,\,3\right],\ \frac{1}{2}\left(1+\sqrt{5}\right),\ \frac{1}{4}\sqrt{26+\frac{58}{\sqrt{5}}}\right\}\right\}$$

As we see in the next cell that the pentagonal base is located in the plane $z = -\sqrt{\frac{25}{8}+\frac{41}{8\sqrt{5}}}$.

In[10]:=
```
Min[vcor〚All, 3〛]
```

Out[10]=
$$-\sqrt{\frac{25}{8}+\frac{41}{8\sqrt{5}}}$$

Here are the coordinates of the polyhedron's base (sorted into counter clockwise order).

In[11]:=
$$\text{base} = \left\{\left\{-\frac{1}{4}\sqrt{2-\frac{2}{\sqrt{5}}},\ \frac{1}{1-\sqrt{5}},\ -\sqrt{\frac{25}{8}+\frac{41}{8\sqrt{5}}}\right\},\ \left\{\sqrt{\frac{1}{4}+\frac{1}{2\sqrt{5}}},\ -\frac{1}{2},\ -\sqrt{\frac{25}{8}+\frac{41}{8\sqrt{5}}}\right\},\ \left\{\sqrt{\frac{1}{4}+\frac{1}{2\sqrt{5}}},\ \frac{1}{2},\ -\sqrt{\frac{25}{8}+\frac{41}{8\sqrt{5}}}\right\},\ \left\{-\frac{1}{4}\sqrt{2-\frac{2}{\sqrt{5}}},\ \frac{1}{4}\left(1+\sqrt{5}\right),\ -\sqrt{\frac{25}{8}+\frac{41}{8\sqrt{5}}}\right\},\ \left\{\text{Root}\left[1-5\,\#1^2+5\,\#1^4\,\&,\,1\right],\ 0,\ -\sqrt{\frac{25}{8}+\frac{41}{8\sqrt{5}}}\right\}\right\};$$

Numerically this plane of the pentagonal base represents the table top at $z \approx -2.32744$ prior to titling the polyhedron.

In[12]:=
```
Min[vcor〚All, 3〛] // N
```

Out[12]= -2.32744

Finding The Optimal Tilt Direction

We first want to determine in what direction the table must be tipped to give rise to the smallest maximum angle beyond which the polyhedron will fall.

Note that one of the edges on the pentagonal base is perpendicular to the x-axis and lies in the plane $x = \sqrt{\frac{1}{4}+\frac{1}{2\sqrt{5}}}$. Due to the symmetry of the polyhedron we only need to consider tilting the table about a point lying along *one-half* of this edge which is defined by the following line segment:

$$\left\{\left\{\sqrt{\frac{1}{4}+\frac{1}{2\sqrt{5}}},\ -\frac{1}{2},\ -\sqrt{\frac{25}{8}+\frac{41}{8\sqrt{5}}}\right\},\ \left\{\sqrt{\frac{1}{4}+\frac{1}{2\sqrt{5}}},\ \frac{1}{2},\ -\sqrt{\frac{25}{8}+\frac{41}{8\sqrt{5}}}\right\}\right\}$$

We define this tipping point parametrically in terms of *t* where $0 \le t \le 1$.

$$\left\{\sqrt{\frac{1}{4}+\frac{1}{2\sqrt{5}}},\ \frac{t}{2},\ -\sqrt{\frac{25}{8}+\frac{41}{8\sqrt{5}}}\right\}$$

We require a vector which is perpendicular to a line from $\left\{0, 0, -\sqrt{\frac{25}{8}+\frac{41}{8\sqrt{5}}}\right\}$ to this tipping point about which the polyhedron will be rotated through some angle (this being the same as tipping the table top). We can find this vector knowing that it must produce a zero dot product.

In[13]:= $\texttt{Solve}\left[\left\{\sqrt{\frac{1}{4}+\frac{1}{2\sqrt{5}}},\ \frac{t}{2}\right\}.\{x, 1\} == 0, x\right]$

Out[13]= $\left\{\left\{x \to -\sqrt{\frac{5}{5+2\sqrt{5}}}\ t\right\}\right\}$

We test that the vector we found produces a zero dot product.

In[14]:= $\left\{\sqrt{\frac{1}{4}+\frac{1}{2\sqrt{5}}},\ \frac{t}{2}, 0\right\}.\left\{-\sqrt{\frac{5}{5+2\sqrt{5}}}\ t, 1, 0\right\}$ // Simplify

Out[14]= 0

So we can now define a rotation transform to carry out the tipping of the polyhedron through an angle of ϕ around this vector.

In[15]:= $r = \texttt{RotationTransform}\left[\phi, \left\{-\sqrt{\frac{5}{5+2\sqrt{5}}}\ t, 1, 0\right\}, \left\{\sqrt{\frac{1}{4}+\frac{1}{2\sqrt{5}}},\ \frac{t}{2}, -\sqrt{\frac{25}{8}+\frac{41}{8\sqrt{5}}}\right\}\right];$

The list of vertices defining each edge of the polyhedron is as follows:

```
In[16]:= edges = PolyhedronData["TruncatedIcosahedron", "EdgeIndices"];
```

The vertex coordinates after tipping are developed by applying the rotation transform in the next cell.

```
In[17]:= vcor = r[N[vcor]];
```

Now we can visualize the situation. The polyhedron is being tipped about the blue point at one extreme of an edge in the green pentagonal base (the other extreme being the center point of this edge). The polyhedron is tipped around the blue vector which lies in the plane of the table top and passes through the tipping point. If the centroid of the liquid inside the polyhedron lies further from the center of the green pentagonal base at $\left\{0, 0, -\sqrt{\frac{25}{8}+\frac{41}{8\sqrt{5}}}\right\}$ than the blue tipping point (as measured in the plane $z = -\sqrt{\frac{25}{8}+\frac{41}{8\sqrt{5}}}$) the polyhedron will fall over.

```
In[18]:= Block[{t = 1., ϕ = 0.282372, z = 0.95},
  Show[
   Graphics3D[{Line /@ (vcor[[#]] & /@ edges), Red, PointSize[Large], Point[vcor], Green,
     Polygon[base], Blue, PointSize[0.03], Point[{Sqrt[1/4 + 1/(2 Sqrt[5])], t/2, -Sqrt[25/8 + 41/(8 Sqrt[5])]}],
     Thick, Arrowheads[0.03], Arrow[{{Sqrt[1/4 + 1/(2 Sqrt[5])], t/2, -Sqrt[25/8 + 41/(8 Sqrt[5])]},
       {1/2 Sqrt[1 + 2/Sqrt[5]] - Sqrt[5 - 2 Sqrt[5]] t, (2 + t)/2, -Sqrt[25/8 + 41/(8 Sqrt[5])]}}], Orange, Opacity[0.35],
     InfinitePlane[r[{{0, 0, -Sqrt[25/8 + 41/(8 Sqrt[5])]}, {0, 1, -Sqrt[25/8 + 41/(8 Sqrt[5])]},
        {-1, 2, -Sqrt[25/8 + 41/(8 Sqrt[5])]}}]]}], Axes → True, ImageSize → 480]]
```

Out[18]=



Looking at the geometry of this visualization we conjecture that the angle we are seeking is realized when the blue point at which the polyhedron is tipped is located at the mid-point of an edge of the green pentagonal base. We'll analyze this conjecture in what follows.

Next we write a function to find the points defining the convex hull of the liquid inside the tipped polyhedron. The hull consists of those points which are the vertices of the polyhedron lying below the plane parallel to the floor which cuts through the polyhedron forming the liquid's surface. Recall that the surface of the liquid must always be parallel to the floor even though the table top is tipped.

In[19]:=
```
Clear[hull];
hull[zplane_?NumericQ, θval_?NumericQ, tval_?NumericQ] :=
  Module[{xyz, z, epairs, zscale, newpts, xypruned},
   (* this function uses global variables vcor and edges *)
   xyz = vcor /. {ϕ → Max[θval, 10.^-99], t → Max[tval, 10.^-99]};
   (* adust for problems if zplane=0 *)
   z = If[zplane == 0.0, 10.^-100, zplane];
   (* get pairs of points composing each edge, lowest z sorted first *)
   epairs = Sort[#, #1[[3]] < #2[[3]] &] & /@ (xyz[[#]] & /@ edges);
   (* get a 0-1 scaled value of z for cutting plane *)
   zscale = Quiet@Rescale[z, #] & /@ epairs[[All, All, 3]];
   (* zero out any results which are not in 0-1 interval *)
   zscale = Map[If[# > 1. || # < 0., 0., #] &, zscale /. ComplexInfinity → 10.^99];
   (* computes points on the
    convex hull where the z plane cuts the polyhedron *)
   newpts = DeleteCases[zscale (epairs[[All, 2]] - epairs[[All, 1]]), {0., 0., 0.}] +
     epairs[[All, 1]][[Flatten[Position[Positive[zscale], True]]]];
   (* delete any points above the z cutting plane *)
   xypruned = DeleteCases[xyz, {xcor_, ycor_, zcor_} /; zcor > z];
   (* output points needed to find the convex hull mesh *)
   Join[xypruned, newpts]];
```

Now we can visualize the tipped polyhedron in the same configuration shown above, but with some liquid inside.

In[21]:=
```
Block[{t = 1., ϕ = 0.282372, z = 0.95},
 Show[Graphics3D[{Line /@ (vcor[[#]] & /@ edges), Red, PointSize[Large],
    Point[vcor], Green, Polygon[base], Blue, PointSize[0.03],
```

$$\text{Point}\left[\left\{\sqrt{\frac{1}{4}+\frac{1}{2\sqrt{5}}},\ \frac{t}{2},\ -\sqrt{\frac{25}{8}+\frac{41}{8\sqrt{5}}}\right\}\right],\ \text{Thick},\ \text{Arrowheads}[0.03],$$

$$\text{Arrow}\left[\left\{\left\{\sqrt{\frac{1}{4}+\frac{1}{2\sqrt{5}}},\ \frac{t}{2},\ -\sqrt{\frac{25}{8}+\frac{41}{8\sqrt{5}}}\right\},\ \left\{\frac{1}{2}\sqrt{1+\frac{2}{\sqrt{5}}}-\sqrt{5-2\sqrt{5}}\ t,\right.\right.\right.$$

$$\left.\left.\left.\frac{2+t}{2},\ -\sqrt{\frac{25}{8}+\frac{41}{8\sqrt{5}}}\right\}\right\}\right],\ \text{Orange},\ \text{Opacity}[0.35],\ \text{InfinitePlane}\Big[$$

$$r\left[\left\{\left\{0,0,-\sqrt{\frac{25}{8}+\frac{41}{8\sqrt{5}}}\right\},\ \left\{0,1,-\sqrt{\frac{25}{8}+\frac{41}{8\sqrt{5}}}\right\},\ \left\{-1,2,-\sqrt{\frac{25}{8}+\frac{41}{8\sqrt{5}}}\right\}\right\}\right]\Big]\}],$$

```
  ConvexHullMesh[hull[z, ϕ, t]] , Axes → True, ImageSize → 480]]
```

Out[21]=



When $t = 0$ the maximum angle through which the polyhedron can be tipped is given by the dihedral angle of the polyhedron's base.

In[22]:= `π - PolyhedronData["TruncatedIcosahedron", "DihedralAngles"] // Min`

Out[22]= `π - ArcCos[Root[1 - 30 #1^2 + 45 #1^4 &, 1]]`

In[23]:= `N[%]`

Out[23]= `0.652358`

In degrees this angle is:

In[24]:= `% / Degree`

Out[24]= `37.3774`

The next function calculates the $\{x, y\}$ coordinates of the centroid of the mesh formed by the liquid. It then uses these coordinates to find the distance, in the plane of the green pentagon, of this centroid from the point about which the polyhedron is being tipped. To find the angle we seek we want this distance to be zero. Note that here we use a Delaunay mesh for the liquid as it seems to give results having good numerical stability.

In[25]:=
```
Clear[cdist];
cdist[z_?NumericQ, φ_?NumericQ, t_?NumericQ] := EuclideanDistance[
  Chop[RegionCentroid[DelaunayMesh[hull[z, φ, t]]][[1 ;; 2]]], {Sqrt[1/4 + 1/(2 Sqrt[5])], t/2}];
```

$$\texttt{Chop[RegionCentroid[DelaunayMesh[hull[z, }\varphi\texttt{, t]]][[1 ;; 2]]], }\left\{\sqrt{\frac{1}{4}+\frac{1}{2\sqrt{5}}}, \frac{t}{2}\right\}\texttt{];}$$

Now we can find the maximum angle to which the polyhedron can be tipped for various values of the parameter $0 \le t \le 1$ and the height of the plane of the liquid's surface $-1.75 \le z \le 1.75$. Here we do this for $t = 0$.

In[27]:=
```
Block[{t = Max[0.0, 10.^-99]},
  pts1 = ParallelTable[
    {z, ϕ /. Quiet[FindRoot[cdist[z, ϕ, t], {ϕ, 0.3}, AccuracyGoal → 4]]},
    {z, -1.75, 1.75, 0.1}]];
```

We plot these points and we see that for $t = 0$ the angle ϕ appears to reach a minimum value near the point $z = -1.25$.

In[28]:=
```
ListLinePlot[pts1, Mesh → 50, MeshStyle → {Red, PointSize[0.01]},
 PlotRange → All, AxesLabel → {z, ϕ}, ImageSize → 480]
```

Out[28]=



We'll continue these calculations for $t = \{0.125, 0.250, 0.750, 1.00\}$.

In[29]:=
```
Block[{t = 0.125},
  pts2 = ParallelTable[
    {z, ϕ /. Quiet[FindRoot[cdist[z, ϕ, t], {ϕ, 0.3}, AccuracyGoal → 4]]},
    {z, -1.75, 1.75, 0.1}]];
```

In[30]:=
```
Block[{t = 0.25},
  pts3 = ParallelTable[
    {z, ϕ /. Quiet[FindRoot[cdist[z, ϕ, t], {ϕ, 0.3}, AccuracyGoal → 4]]},
    {z, -1.75, 1.75, 0.1}]];
```

In[31]:=
```
Block[{t = 0.5},
  pts4 = ParallelTable[
    {z, ϕ /. Quiet[FindRoot[cdist[z, ϕ, t], {ϕ, 0.3}, AccuracyGoal → 4]]},
    {z, -1.75, 1.75, 0.1}]];
```

In[32]:=
```
Block[{t = 0.75},
  pts5 = ParallelTable[
    {z, ϕ /. Quiet[FindRoot[cdist[z, ϕ, t], {ϕ, 0.3}, AccuracyGoal → 4]]},
    {z, -1.75, 1.75, 0.1}]];
```

In[33]:=
```
Block[{t = 1.0},
  pts6 = ParallelTable[
    {z, ϕ /. Quiet[FindRoot[cdist[z, ϕ, t], {ϕ, 0.3}, AccuracyGoal → 4]]},
    {z, -1.75, 1.75, 0.1}]];
```

We plot all of our results in the next cell.

In[34]:=
```
ListLinePlot[
 {Tooltip[pts1, "t=0.000"], Tooltip[pts2, "t=0.125"], Tooltip[pts3, "t=0.250"],
  Tooltip[pts4, "t=0.500"], Tooltip[pts5, "t=0.750"], Tooltip[pts6, "t=1.000"]},
 InterpolationOrder → 3, PlotRange → All, AxesLabel → {z, ϕ}, ImageSize → 480]
```

Out[34]=



The bottom line in this graph corresponds to $t = 0$ and each line higher up in the graph corresponds to an ever increasing value of t. From this analysis we can see that our conjecture was correct–when the table is tipped in the direction of a mid-point of an edge of the base the angle to which the polyhedron may ever be tipped without falling is minimized. We'll use this conclusion in the next section to find the tip angle and the liquid's volume.

Calculating The Angle And Volume

With the conclusion reached in the prior section that the angle we seek is achieved when $t = 0$ (i.e., the tip angle arises when the polyhedron in tipped in the direction of a mid-point along an edge of its base) we can greatly simplify our calculations.

We define a new rotation transform in which $t = 0$ at all times.

In[35]:=
```
r2 = RotationTransform[ϕ, {0, 1, 0}, {Sqrt[1/4 + 1/(2 Sqrt[5])], 0, -Sqrt[25/8 + 41/(8 Sqrt[5])]}];
```

We rotate the vertex coordinates using this new transformation.

In[36]:=
```
vcor = r2[N[PolyhedronData["TruncatedIcosahedron", "VertexCoordinates"]]];
```

We modify our convex hull function to eliminate the parameter *t*.

In[37]:=
```
Clear[hull2];
hull2[zplane_?NumericQ, θ_?NumericQ] :=
  Module[{xyz, z, epairs, zscale, newpts, xypruned},
   (* this function uses global variables vcor and edges *)
   xyz = vcor /. ϕ → θ;
   (* adust for problems if zplane=0 *)
   z = If[zplane == 0.0, 10.^-100, zplane];
   (* get pairs of points composing each edge, lowest z sorted first *)
   epairs = Sort[#, #1[[3]] < #2[[3]] &] & /@ (xyz[[#]] & /@ edges);
   (* get a 0-1 scaled value of z for cutting plane *)
   zscale = Quiet@Rescale[z, #] & /@ epairs[[All, All, 3]];
   (* zero out any results which are not in 0-1 interval *)
   zscale = Map[If[# > 1. || # < 0., 0., #] &, zscale /. ComplexInfinity → 10.^99];
   (* computes points on the
    convex hull where the z plane cuts the polyhedron *)
   newpts = DeleteCases[zscale (epairs[[All, 2]] - epairs[[All, 1]]), {0., 0., 0.}] +
     epairs[[All, 1]][[Flatten[Position[Positive[zscale], True]]]];
   (* delete any points above the z cutting plane *)
   xypruned = DeleteCases[xyz, {xcor_, ycor_, zcor_} /; zcor > z];
   (* output points needed to find the convex hull mesh *)
   Join[xypruned, newpts]];
```

We write a new function to use this simplified data. It computes the *x* coordinate of the centroid of the Delaunay mesh formed by the liquid inside the polyhedron.

In[39]:=
```
Clear[centroid];
centroid[z_?NumericQ, φ_?NumericQ] :=
  RegionCentroid[DelaunayMesh[hull2[z, φ]]][[1]];
```

We can quickly calculate the maximum allowable tip angle over the range $-2.3 \leq z \leq 2.3$.

In[41]:=
```
pts = ParallelTable[
    {z, φ /. FindRoot[centroid[z, φ] == Sqrt[1/4 + 1/(2 Sqrt[5])], {φ, 0.3}]}, {z, -2.3, 2.3, 0.005}];
```

We will also determine the maximum angle to which the polyhedron may be tipped without falling if it is completely filled with liquid.

In[42]:=
```
FindRoot[r2[{0, 0, 0}][[1]] == Sqrt[1/4 + 1/(2 Sqrt[5])], {ϕ, 0.3}]
```

Out[42]=
```
{ϕ → 0.287494}
```

We plot these data and see that the maximum angle to which the polyhedron can be safely titled occurs at the minimum on this curve near the point $z = -1.26$. The green line parallel to the *x*-axis indicates the angle, found in the prior cell, beyond which a truncated icosahedron which is completely filled with liquid would fall.

In[43]:=
```
Show[ListLinePlot[pts, Mesh → 60, MeshStyle → {Red, PointSize[0.01]},
  PlotRange → All, AxesLabel → {z, ϕ}, ImageSize → 480], Graphics[
  {Thick, Darker[Green, 0.35], Line[{{-2.35, 0.287494}, {2.35, 0.287494}}]}]]
```

Out[43]=



Now we can find this minimum angle and the corresponding value of *z* giving the height of the plane of the liquid's surface.

In[44]:= $$\texttt{sol = FindMinimum}\left[\left\{\phi,\ \texttt{centroid}[z,\phi] == \sqrt{\frac{1}{4}+\frac{1}{2\sqrt{5}}}\right\},\ \{\{\phi, 0.282\}, \{z, -1.26\}\}\right]$$

Out[44]= {0.282372, {ϕ → 0.282372, z → −1.25324}}

In degrees, the maximum tilt angle of the table so that the canister will never lose its balance, independent of the direction of the tilt and the filling level of the canister, is found to be:

In[45]:=
```
(ϕ /. sol[[2]]) / Degree
```

Out[45]= 16.1787

If this maximum tilt angle is marginally exceeded and the canister starts to tumble, the fraction of its volume that is filled with fluid is found to be:

In[46]:= $$\frac{\texttt{Volume[DelaunayMesh[hull2[z, }\phi\texttt{] /. sol[[2]]]]}}{\texttt{PolyhedronData["TruncatedIcosahedron", "Volume"]}}$$

Out[46]= 0.117396

We visualize this optimal solution in the next cell.

```
In[47]:= Block[{t = 10.^-99, ϕ = 0.282372, z = -1.25324, vcor},
  vcor = r[N[PolyhedronData["TruncatedIcosahedron", "VertexCoordinates"]]];
  Show[Graphics3D[{Line /@ (vcor[[#]] & /@ edges), Red, PointSize[Large],
     Point[vcor], Green, Polygon[base], Blue, PointSize[0.03],
     Point[{Sqrt[1/4 + 1/(2 Sqrt[5])], t/2, -Sqrt[25/8 + 41/(8 Sqrt[5])]}], Thick, Arrowheads[0.03],
     Arrow[{{Sqrt[1/4 + 1/(2 Sqrt[5])], t/2, -Sqrt[25/8 + 41/(8 Sqrt[5])]}, {1/2 Sqrt[1 + 2/Sqrt[5]] - Sqrt[5 - 2 Sqrt[5]] t,
        (2 + t)/2, -Sqrt[25/8 + 41/(8 Sqrt[5])]}}], Orange, Opacity[0.35], InfinitePlane[
      r[{{0, 0, -Sqrt[25/8 + 41/(8 Sqrt[5])]}, {0, 1, -Sqrt[25/8 + 41/(8 Sqrt[5])]}, {-1, 2, -Sqrt[25/8 + 41/(8 Sqrt[5])]}}]]}],
   ConvexHullMesh[hull[z, ϕ, t]], Boxed → False, ViewPoint → {0.34, -3.36, 0.11},
   ImageSize → Large]]
```

Out[47]=



Here is a visualization of the slope of the table top at a tilt angle of 16.1787°. It would seem that the canister is going to be fairly stable.

In[48]:=
```
Plot[(5 - x) Tan[0.282372], {x, 0, 5}, PlotTheme → "Marketing",
 Prolog → {GrayLevel[0, 0.3], Rectangle[Scaled[{0, 0}], Scaled[{1, 1}]]},
 AspectRatio -> Automatic, ImageSize → Large]
```

Out[48]=